Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Test Statistics



NOTICE

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁑ . When you see ⁑ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01◆LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL^{T}SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02^{T}THIS IS A*
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	*03^T ⊢ SAMPLE*
04 AVIEW	■ AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↗*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] ■ [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	*11^{T}R3=*
12 ARCL 03	■ [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	■ [RTN]	*14 RTN*

TABLE OF CONTENTS

ONE SAMPLE TEST STATISTICS FOR THE MEAN

Suppose $\{x_1, x_2, \ldots, x_n\}$ is a sample from a normal population with a known variance σ^2 and unknown mean μ. A test of the null hypothesis

$$H_0 : \mu = \mu_0$$

is based on the z statistic which has a standard normal distribution.

If the variance σ^2 is unknown then the t statistic, which has the t distribution with n - 1 degrees of freedom, is used instead.

Equations:

$$z = \frac{\sqrt{n}\,(\bar{x} - \mu_0)}{\sigma}$$

$$t = \frac{\sqrt{n}\,(\bar{x} - \mu_0)}{s}$$

where $\bar{x}$ and s are sample mean and sample standard deviation.

Remark: $n > 1$.

Reference: This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Calculate the z and the t statistics for the following set of data if $\mu_0 = 2$ and $\sigma = 1$.

{2.73, 0.45, 2.52, 1.19, 3.51}

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] ONEST [ALPHA]	ONE SAMPLE T.
2.73 [Σ+] .45 [Σ+] 2.52 [Σ+]	
1.19 [Σ+] 3.51 [Σ+]	5.00
[R/S]	MU NAUGHT ?
2 [R/S]	SIGMA ?
1 [R/S]	Z=0.18
[R/S]	T=0.14
[R/S]	XBAR=2.08
[R/S]	S=1.24

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program.			
2	Initialize the program.		[XEQ] ONEST	ONE SAMPLE T.
3	Input data. Repeat steps 3-4 for			
	$i = 1, 2, \ldots, n$.	x_i	[Σ+]	(i)
4	If you make a mistake inputting x_k, delete			
	it and go to step 3.	x_k as entered	[Σ-]	(k-1)
5	Input μ_0 and σ and calculate z and t.		[R/S]	MU NAUGHT ?
		μ_0	[R/S]	SIGMA ?
		σ	[R/S]	Z=(z)
			[R/S]	T=(t)
			[R/S]	XBAR=($\bar{x}$)
			[R/S]	S=(s)
6	To calculate z and t for a different pair			
	of μ_0 and σ, go to step 5.			
7	To use the program for another set of			
	data, go to step 2.			

Program Listings

```
 01♦LBL "ONE
ST"
 02 FIX 2
 03 CLRG
 04 ΣREG 00
 05 "ONE SAM
PLE T."
 06 AVIEW
 07 STOP
```

Initialize

```
 08♦LBL E
 09 "MU NAUG
HT ?"
 10 PROMPT
 11 STO 06
 12 "SIGMA ?
"
 13 PROMPT
 14 STO 07
 15 MEAN
 16 RCL 06
 17 -
 18 RCL 05
 19 SQRT
 20 *
 21 STO 08
 22 RCL 07
 23 /
 24 "Z"
 25 XEQ 11
 26 SDEV
 27 RCL 08
 28 X<>Y
 29 /
 30 "T"
 31 XEQ 11
 32 MEAN
 33 "XBAR"
 34 XEQ 11
 35 SDEV
 36 "S"
 37 XEQ 11
 38 XEQ E
```

Store μ_0 and σ and make calculations

```
 39♦LBL 11
 40 "⊢="
 41 ARCL X
 42 AVIEW
 43 STOP
 44 RTN
 45 .END.
```

Display subroutine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Σx	50	
	Σx^2		
	Σy		
	Σy^2		
	Σxy		
05	n	55	
	μ_0		
	σ		
	$\sqrt{n}(x-\mu_0)$		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 24 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ONE SAMPLE TEST
STATISTICS FOR THE MEAN
PROGRAM REGISTERS NEEDED: 16

ROW 1 (1 - 4)

ROW 2 (4 - 5)

ROW 3 (5 - 9)

ROW 4 (9 - 12)

ROW 5 (12 - 23)

ROW 6 (24 - 31)

ROW 7 (31 - 36)

ROW 8 (37 - 42)

ROW 9 (43 - 45)

TEST STATISTICS FOR THE CORRELATION COEFFICIENT

Under the assumptions of normal correlation analysis, the t statistic , which has the t distribution with n - 2 degrees of freedom, can be used to test the null hypothesis that the true correlation coefficient $\rho = 0$.

To test the null hypothesis $\rho = \rho_0$, where ρ_0 is a given number, the z statistic is used. z has approximately the standard normal distribution.

Equations:

$$t = \frac{r\sqrt{n-2}}{\sqrt{1-r^2}}$$

$$z = \frac{\sqrt{n-3}}{2} \ln\left[\frac{(1+r)(1-\rho_0)}{(1-r)(1+\rho_0)}\right]$$

where r is an estimate (based on a sample of size n) of the correlation coefficient ρ.

Remarks: 1. This program requires that $n > 3$, $|r| < 1$ and $|\rho_0| < 1$; otherwise "DATA ERROR" will result.

2. Usually, the z statistic is used when the sample size is large.

References:

1. Hogg and Craig, Introduction to Mathematical Statistics, Macmillan and Co., 1970.
2. J. Freund, Mathematical Statistics, Prentice-Hall, 1971.
3. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Given r = 0.12, n = 31, and $\rho_0 = 0$, find t and z.

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 003	
[XEQ] [ALPHA] CORRTS [ALPHA]	COR. COEF. T.S.
	N ?
31 [R/S]	R ?
.12 [R/S]	T=0.65
[E]	RHO NAUGHT ?
0 [R/S]	Z=0.64

User Instructions

SIZE: 003

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] CORRTS	COR. COEF. T.S.
3	To calculate t,			N ?
		n	[R/S]	R ?
		r	[R/S]	T=
4	To calculate z,		[E]	RHO NAUGHT ?
		ρ_0	[R/S]	Z=
5	For a new case, go to step 3 or 4.			

Program Listings

Initialize

```
01◆LBL "COR
RTS"
02 FIX 2
03 "COR. CO
EF. T.S."
04 AVIEW
05 PSE
```

n

Test n > 3?

r

Test |r| < 1?

```
06 "N ?"
07 PROMPT
08 STO 01
09 3
10 X<>Y
11 X<=Y?
12 GTO 09
13 "R ?"
14 PROMPT
15 STO 00
16 XEQ 00
```

Calculate t

```
17 RCL 01
18 2
19 -
20 1
21 RCL 00
22 X↑2
23 -
24 /
25 SQRT
26 RCL 00
27 *
28 "T"
29 GTO 11
```

Test $|\rho_0|$ < 1

Calculate z

```
30◆LBL E
31 "RHO NAU
GHT ?"
32 PROMPT
33 STO 02
34 XEQ 00
35 RCL 00
36 1
37 +
38 1
39 RCL 00
40 -
41 /
42 1
43 RCL 02
44 -
45 *
46 1
47 RCL 02
48 +
49 /
50 LN
51 RCL 01
52 3
53 -
54 SQRT
55 *
56 2
57 /
58 "Z"
```

Display routine

```
59◆LBL 11
60 "├="
61 ARCL X
62 AVIEW
63 STOP
64 RTN
```

Test r and ρ_0

```
65◆LBL 00
66 ABS
67 1
68 X<>Y
69 X>Y?
70 GTO 09
71 RTN
```

Generate "DATA ERROR"

```
72◆LBL 09
73 0
74 /
75 .END.
```

80	
90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	r	50	
	n		
	ρ_0		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 003 TOT. REG. 21 USER MODE
ENG ____ FIX 2 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

TEST STATISTICS FOR
THE CORRELATION COEFFICIENT
PROGRAM REGISTERS NEEDED: 19

ROW 1 (1 - 3)

ROW 2 (3 - 3)

ROW 3 (3 - 11)

ROW 4 (12 - 18)

ROW 5 (19 - 29)

ROW 6 (30 - 31)

ROW 7 (31 - 40)

ROW 8 (41 - 53)

ROW 9 (54 - 62)

ROW 10 (63 - 74)

ROW 11 (75 - 75)

DIFFERENCES AMONG PROPORTIONS

Suppose x_1, x_2, ..., x_k are observed values of a set of independent random variables having binomial distributions with parameters n_i and θ_i ($i = 1, 2, \ldots, k$).

A chi-square statistic χ^2 can be used to test the null hypothesis $\theta_1 = \theta_1 = \ldots = \theta_k$. The χ^2 statistic has the chi-square distribution with $k - 1$ degrees of freedom.

Equations:

$$\chi^2 = \sum_{i=1}^{k} \frac{(x_i - n_i\hat{\theta})^2}{n_i\hat{\theta}(1 - \hat{\theta})} = \sum_{i=1}^{k} n_i \left[\frac{1}{\sum_{i=1}^{k} x_i} \sum_{i=1}^{k} \frac{x_i^2}{n_i} + \frac{1}{\sum_{i=1}^{k} (n_i - x_i)} \sum_{i=1}^{k} \frac{(n_i - x_i)^2}{n_i} - 1 \right]$$

where

$$\hat{\theta} = \sum_{i=1}^{k} x_i \Big/ \sum_{i=1}^{k} n_i$$

References: 1. J. Freund, Mathematical Statistics, Prentice-Hall, 1971.
2. This program is a translation of the HP-65 State Pac 2 program.

Example:

	n_i	x_i
Sample 1	400	232
Sample 2	500	260
Sample 3	400	197

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] DIFF [ALPHA]	DIFF. A. PROPS
	N1 ?
400 [R/S]	X1 ?
232 [R/S]	N2 ?
500 [R/S]	X2 ?
260 [R/S]	N3 ?
400 [R/S]	X3 ?
197 [R/S]	N4 ?
[E]	CHI-SQ=6.47
[R/S]	dF=2.00
[R/S]	THETA=0.53

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] DIFF	DIFF. A. PROPS
3	Input data. Repeat steps 3-4 for			N1 ?
	i = 1, 2, ..., n.	n_i	[R/S]	X(i)?
		x_i	[R/S]	N(i+1)?
4	If you make a mistake inputtine n_k or x_k,		[C]	N(K)?
	delete the incorrect entry and go back to	n_k as entered	[R/S]	X(K)?
	step 3.	x_k as entered	[R/S]	N(K)?
5	Calculate χ^2.		[E]	CHI-SQ=(χ^2)
6	Calculate df.		[R/S]	dF=(df)
7	Calculate $\hat{\theta}$.		[R/S]	THETA=($\hat{\theta}$)
8	To use the program for another set of data,			
	go to step 2.			

Program Listings

01♦LBL "DIF F"	
02 FIX 2	
03 CLRG	Initialize
04 CF 00	
05 CF 29	
06 "DIFF. A PROPS"	
07 AVIEW	
08 PSE	
09 GTO A	
10♦LBL C	For corrections
11 SF 00	
12♦LBL A	
13 1	
14 FS? 00	
15 CHS	
16 ST+ 03	
17 "N"	
18 XEQ 12	
19 STO 07	n_i
20 "X"	
21 XEQ 12	
22 STO 08	x_i
23 FS? 00	
24 CHS	
25 ST+ 01	
26 ABS	
27 -	
28 STO 04	
29 FS? 00	accumulate sums
30 CHS	
31 ST+ 02	
32 ABS	
33 RCL 08	
34 +	
35 STO 09	
36 RCL 08	
37 X↑2	
38 X<>Y	
39 /	
40 FS? 00	
41 CHS	
42 ST+ 05	
43 ABS	
44 RCL 04	
45 X↑2	
46 RCL 09	
47 /	
48 FS? 00	
49 CHS	

50 ST+ 06	
51 FC?C 00	
52 GTO A	
53 1	
54 ST- 03	
55 GTO A	
56♦LBL E	
57 RCL 05	
58 RCL 01	Calculate y^2
59 /	
60 RCL 06	
61 RCL 02	
62 /	
63 +	
64 1	
65 -	
66 RCL 01	
67 RCL 02	
68 +	
69 *	
70 "CHI-SQ"	
71 XEQ 11	
72 RCL 03	Calculate df
73 2	
74 -	
75 "dF"	
76 XEQ 11	
77 RCL 01	
78 RCL 01	
79 RCL 02	Calculate $\hat{\theta}$
80 +	
81 /	
82 "THETA"	
83♦LBL 11	
84 "⊢="	
85 ARCL X	Display result routine
86 AVIEW	
87 STOP	
88 RTN	
89♦LBL 12	
90 FIX 0	
91 "⊢"	
92 ARCL 03	Display input routine
93 "⊢ ?"	
94 AVIEW	
95 FIX 2	
96 STOP	
97 RTN	
98 .END.	

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	Σx_i		
	$\Sigma\ (n_i - x_i)$		
	k		
	$n_i - x_i$		
05	$\Sigma\ (x_i^2/n_i)$	55	
	$\Sigma\ (n_i - x_i)^2/n_i$		
	n_i		
	x_i		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 35 USER MODE
ENG ___ FIX 2 SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		For correction	Normal
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

DIFFERENCES AMONG PROPORTIONS

PROGRAM REGISTERS NEEDED: 26

ROW 1 (1 - 4)

ROW 2 (5 - 6)

ROW 3 (6 - 11)

ROW 4 (12 - 18)

ROW 5 (19 - 26)

ROW 6 (27 - 37)

ROW 7 (38 - 48)

ROW 8 (48 - 55)

ROW 9 (55 - 65)

ROW 10 (66 - 71)

ROW 11 (71 - 79)

ROW 12 (80 - 85)

ROW 13 (85 - 93)

RCW 14 (93 - 98)

BEHRENS-FISHER STATISTIC

Suppose $\{x_1, x_2, \ldots, x_{n_1}\}$ and $\{y_1, y_2, \ldots, y_{n_2}\}$ are independent random samples from two normal populations having means μ_1, μ_2 (unknown). If the variances σ_1^2, σ_2^2 cannot be assumed equal, then the Behrens-Fisher statistic d is used instead of the t statistic to test the null hypothesis

$$H_0 : \mu_1 - \mu_2 = D$$

Equation:

$$d = \frac{\bar{x} - \bar{y} - D}{\sqrt{\dfrac{s_1^2}{n_1} + \dfrac{s_2^2}{n_2}}}$$

where $\bar{x}$, $\bar{y}$ and s_1^2, s_2^2 are sample means and variances.

Critical values of this test are tabulated in the Fisher-Yates Tables for various values of n_1, n_2, α and θ, where α is the level of significance and

$$\theta = \tan^{-1}\left(\frac{s_1}{s_2}\sqrt{\frac{n_2}{n_1}}\right)$$

Remark: $n_1 > 1$, $n_2 > 1$.

References:
1. Fisher and Yates, Statistical Tables for Biological, Agricultural and Medical Research, Hafner, Publishing Co., 1970.
2. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Calculate the Behrens-Fisher statistic for D = 0.

x: 79, 84, 108
y: 91, 103, 90, 113, 108

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] BEH [ALPHA]	BEHRENS-FISH.
79 [Σ+] 84 [Σ+] 108 [Σ+]	3.00
[R/S]	XBAR=90.33
[R/S]	S2/N=80.11
91 [Σ+] 103 [Σ+] 90 [Σ+] 113 [Σ+]	
108 [Σ+]	5.00
[R/S]	YBAR=101.00
[R/S]	S2/N=20.90
[E]	D ?
0 [R/S]	d=-1.06
[R/S]	THETA=62.94

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] BEH	BEHRENS-FISH.
3	Input x data. Repeat steps 3-4 for			
	$i = 1, 2, \ldots, n_1$.	x_i	[Σ+]	(i)
4	If you make a mistake inputting x_k, delete			
	it and go to step 3.	x_k as entered	[Σ-]	(k-1)
5	Calculate $\bar{x}$.		[R/S]	XBAR=($\bar{x}$)
6	Calculate $s_1{}^2/n$		[R/S]	S2/N=($s_1{}^2/n$)
7	Input y data. Repeat steps 7-8 for			
	$j = 1, 2, \ldots, n_2$.	y_i	[Σ+]	(j)
8	If you make a mistake inputting y_h, delete			
	it and go to step 7.	y_h as entered	[Σ-]	(h-1)
9	Calculate $\bar{y}$.		[R/S]	YBAR=($\bar{y}$)
10	Calculate $s_2{}^2/n$		[R/S]	S2/N=($s_2{}^2/n$)
11	Calculate d.		[E]	D ?
		D	[R/S]	d=(d)
12	Calculate θ.		[R/S]	THETA=(θ)
13	For a different D, go to step 11.			
14	To use the program for other sets of data,			
	go to step 2.			

Program Listings

01♦LBL "BEH" 02 FIX 2 03 CLRG 04 CF 01 05 ΣREG 00 06 "BEHRENS-FISH." 07 AVIEW 08 STOP	Initialize
09♦LBL 05 10 MEAN 11 FS? 01 12 GTO 02 13 STO 06 14♦LBL 02 15 STO 08 16 "XBAR" 17 FS? 01 18 XEQ 01 19 XEQ 11	Calculate and display $\bar{x}$, $\bar{y}$
20 SDEV 21 X↑2 22 RCL 05 23 / 24 FS? 01 25 GTO 03 26 STO 07 27♦LBL 03 28 STO 09 29 CLΣ 30 SF 01 31 "S2/N" 32 XEQ 11 33 GTO 05	Calculate and display s_i^2/N
34♦LBL 01 35 "YBAR" 36 RTN	
37♦LBL E 38 "D ?" 39 PROMPT 40 CHS 41 RCL 08 42 - 43 RCL 06 44 + 45 RCL 07 46 RCL 09 47 + 48 SQRT 49 / 50 "d" 51 XEQ 11	Calculate d
52 RCL 07 53 RCL 09 54 / 55 SQRT 56 ATAN 57 "THETA"	Calculate θ
58♦LBL 11 59 "⊢=" 60 ARCL X 61 CLX 62 AVIEW 63 STOP 64 RTN 65 .END.	Display result routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	used in summations	50	
	used in summations		
	used in summations		
	used in summations		
	used in summations		
05	used in summations	55	
	$\bar{x}$		
	$s_1{}^2/n$		
	$\bar{y}$		
	$s_2{}^2/n$		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 28 USER MODE
ENG ___ FIX 2 SCI ___ ON X OFF ___
DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		"YBAR"	"XBAR"

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BEHRENS–FISHER STATISTIC

PROGRAM REGISTERS NEEDED: 19

ROW 1 (1 - 5)

ROW 2 (5 - 6)

ROW 3 (6 - 15)

ROW 4 (16 - 19)

ROW 5 (20 - 30)

ROW 6 (30 - 35)

ROW 7 (35 - 40)

ROW 8 (41 - 51)

ROW 9 (51 - 58)

ROW 10 (59 - 65)

KRUSKAL-WALLIS STATISTIC

Suppose we want to test the null hypothesis that k independent random samples of sizes $n_1, n_2, \ldots, n_k$ come from identical continuous populations.

Arrange all values from k samples jointly (as if they were one sample) in an increasing order of magnitude. Let R_{ij} $(i = 1, 2, \ldots, k, j = 1, 2, \ldots, n_i)$ be the rank of the jth value in the ith sample.

The Kruskal-Wallis statistic H can be used to test the null hypothesis.

When all sample sizes are large (>5), H is distributed approximately as the chi-square with k - 1 degrees of freedom. For small samples, the test is based on special tables.

Equation:

$$H = \frac{12}{N(N+1)} \sum_{i=1}^{k} \frac{\left(\sum_{j=1}^{n_i} R_{ij}\right)^2}{n_i} - 3(N+1)$$

where

$$N = \sum_{i=1}^{k} n_i$$

References: 1. W.J. Conover, Practical Nonparametric Statistics, John Wiley and Sons, 1971.
2. Table for small samples (k = 3): Alexander and Quade, On the Kruskal-Wallis Three Sample H-statistic, University of North Carolina, Department of Biostatistics, Inst. Statistics Mimeo Ser. 602, 1968.
3. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Ranks R_{ij}

i \ j	1	2	3	4	5	6	7	8	9	10
1	29	5	26	10	33	30				
2	11	12	9	7	20	18	19	21		
3	14	28	8	25	17	15	32	4	2	
4	6	27	3	16	24	13	1	31	22	23

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 006	
[XEQ] [ALPHA] KRU [ALPHA]	KRUSKAL-WALL.
	R1,1 ?
29 [R/S]	R1,2 ?
5 [R/S]	R1,3 ?
26 [R/S]	R1,4 ?
⋮	⋮
30 [R/S]	R1,7 ?
[B]	R2,1 ?
11 [R/S]	R2,2 ?
12 [R/S]	R2,3 ?
⋮	⋮
21 [R/S]	R2,9 ?
[B]	R3,1 ?
14 [R/S]	R3,2 ?
28 [R/S]	R3,3 ?
⋮	⋮
2 [R/S]	R3,10 ?
[B]	R4,1 ?
6 [R/S]	R4,2 ?
27 [R/S]	R4,3 ?
⋮	⋮
23 [R/S]	R4,11 ?
[B]	R5,1 ?
[E]	H=2.29
[R/S]	dF=3.00
[R/S]	N=33.00

User Instructions

				SIZE: 006
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] KRU	KRUSKAL-WALL.
3	Perform steps 3-5 for i = 1, 2, ..., k			R1,1 ?
	and j = 1, 2, ..., n_i. Input R_{ij}.	R_{ij}	[R/S]	R(i),(j+1)?
4	If you make a mistake inputting R_{ih},		[C]	R(i),(h)?
	delete it and go to step 3.	R_{ih} as entered	[R/S]	R(i),(h)?
5	For the end of the i'th sample, press		[B]	R(i+1),1?
6	Calculate H,		[E]	H=
	df,		[R/S]	dF=
	and N		[R/S]	N=
7	To use the program for another set of			
	data, go to step 2.			

Program Listings

Initialize

```
 01♦LBL "KRU
"
 02 CF 29
 03 FIX 0
 04 CLRG
 05 "KRUSKAL
-WALL."
 06 AVIEW
 07 GTO A
```

Correction

```
 08♦LBL C
 09 1
 10 ST- 01
 11 SF 00
```

Input R_{ij}

```
 12♦LBL A
 13 RCL 01
 14 1
 15 +
 16 RCL 04
 17 1
 18 +
 19 "R"
 20 ARCL X
 21 "├,"
 22 ARCL Y
 23 "├ ?"
 24 PROMPT
 25 FS? 00
 26 CHS
 27 ST+ 02
 28 1
 29 FC?C 00
 30 ST+ 01
 31 GTO A
```

Compute row i partial results

```
 32♦LBL B
 33 RCL 01
 34 ST+ 05
 35 RCL 02
 36 X↑2
 37 X<>Y
 38 /
 39 ST+ 03
 40 1
 41 ST+ 04
 42 0
 43 STO 01
 44 STO 02
 45 GTO A
```

Calculate H

```
 46♦LBL E
 47 FIX 2
 48 RCL 03
 49 4
 50 *
 51 RCL 05
 52 /
 53 RCL 05
 54 1
 55 +
 56 /
 57 LASTX
 58 -
 59 3
 60 *
 61 "H"
 62 XEQ 11
```

Calculate df and N

```
 63 RCL 04
 64 1
 65 -
 66 "dF"
 67 XEQ 11
 68 RCL 05
 69 "N"
```

Display routine

```
 70♦LBL 11
 71 "├="
 72 ARCL X
 73 AVIEW
 74 STOP
 75 RTN
 76 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	n_i		
	ΣR_{ij}		
	$\Sigma\ [(\Sigma R_{ij})^2/n_i]$		
	k		
05	N	55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 006 TOT. REG. 25 USER MODE
ENG ___ FIX 2 SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

KRUSKAL–WALLIS STATISTIC

PROGRAM REGISTERS NEEDED: 20

ROW 1 (1 – 5)

ROW 2 (5 – 5)

ROW 3 (6 – 12)

ROW 4 (13 – 21)

ROW 5 (22 – 28)

ROW 6 (29 – 35)

ROW 7 (36 – 45)

ROW 8 (45 – 55)

ROW 9 (56 – 65)

ROW 10 (66 – 71)

ROW 11 (72 – 76)

MEAN SQUARE SUCCESSIVE DIFFERENCE

When test and estimation techniques are used, the method of drawing the sample from the population is specified to be random in most cases. If observations are chosen in sequence $x_1, x_2, \ldots, x_n$, the mean-square successive difference η can be used to test for randomness.

If the sample size n is large (say, greater than 20) and the population is normal, then a z statistic has approximately the standard normal distribution. Long trends are associated with large positive values of z and short oscillations with large negative values.

Equations:

$$\eta = \sum_{i=2}^{n} (x_i - x_{i-1})^2 \bigg/ \sum_{i=1}^{n} (x_i - \bar{x})^2 = \sum_{i=2}^{n} (x_i - x_{i-1})^2 \Bigg/ \left[\sum_{i=1}^{n} x_i^2 - \frac{\left(\sum_{i=1}^{n} x_i \right)^2}{n} \right]$$

$$z = \frac{1 - \eta/2}{\sqrt{\dfrac{n-2}{n^2-1}}}$$

References: 1. Dixon and Massey, Introduction to Statistical Analysis, McGraw-Hill, 1969.
2. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Find the mean-square successive difference for the following set of data:

{0.53, 0.52, 0.39, 0.49, 0.97

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] MNSQD [ALPHA]	MEAN SQ DIFF
.53 [A] .52 [A] .39 [A] .49 [A] .97 [A]	5.00
[E]	ETA=1.27
[R/S]	Z=1.03

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] MNSQD	MEAN SQ DIFF
3	Repeat steps 3-4 for i = 1, 2, ..., n.	x_i	[A]	(i)
4	If you make a mistake inputting x_k, delete			
	it and go to step 3.	x_k	[C]	(k-1)
5	Calculate η		[E]	ETA=(η)
	and z		[R/S]	Z=(z)
6	To use the program for another set of			
	data, go to step 2.			

Program Listings

Program	Comments
01♦LBL "MNSQD" 02 FIX 2 03 CLRG 04 SF 01 05 ΣREG 00 06 "MEAN SQ DIFF" 07 AVIEW 08 STOP	Initialize
09♦LBL C 10 RCL 08 11 STO 07 12 RCL 06 13 - 14 RCL 06 15 Σ- 16 STOP	Correction routine
17♦LBL A 18 STO 06 19 RCL 07 20 STO 08 21 - 22 FS?C 01 23 0 24 RCL 06 25 STO 07 26 Σ+ 27 STOP 28♦LBL E 29 RCL 03 30 RCL 01 31 RCL 00 32 X↑2 33 RCL 05 34 / 35 - 36 / 37 "ETA" 38 XEQ 11	Compute summations $R_y = x_i - x_{i-1}$ Calculate η
39 2 40 / 41 1 42 - 43 CHS 44 RCL 05 45 2 46 - 47 RCL 05 48 X↑2 49 1 50 - 51 / 52 SQRT 53 / 54 "Z"	Calculate z
55♦LBL 11 56 "⊢=" 57 ARCL X 58 AVIEW 59 STOP 60 RTN 61 .END.	Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Σx_i	50	
	Σx_i^2		
	$\Sigma (x_i - x_{i-1})^2$		
05	n	55	
	x_i		
	x_{i-1}		
	used for corrections		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 23 USER MODE
ENG ____ FIX 2 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		for x_1	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

MEAN-SQUARE
SUCCESSIVE DIFFERENCE
PROGRAM REGISTERS NEEDED: 15

ROW 1 (1 - 4)

ROW 2 (4 - 6)

ROW 3 (6 - 15)

ROW 4 (16 - 26)

ROW 5 (27 - 37)

ROW 6 (37 - 46)

ROW 7 (47 - 56)

ROW 8 (57 - 61)

THE RUN TEST FOR RANDOMNESS

Consider a sequence of symbols such that the symbols are of two types only. A run is a continuous string of identical symbols preceded and followed by a different symbol or no symbol at all. For example, the sequence 1110100011 has five runs.

Let the total number of runs in a given sequence be u, and let n_1 and n_2 represent the number of symbols of type 1 and type 2 respectively. If the sample sizes are large (say, n_1 and n_2 are both greater than 10), then the randomness of the sequence may be tested using a z statistic which has the standard normal distribution.

Equations:

The sample distribution of the run has the mean μ and the standard deviation σ.

$$\mu = \frac{2 n_1 n_2}{n_1 + n_2} + 1$$

$$\sigma = \sqrt{\frac{2 n_1 n_2 (2 n_1 n_2 - n_1 - n_2)}{(n_1 + n_2)^2 (n_1 + n_2 - 1)}}$$

The test is based on the statistic

$$z = \frac{u - \mu}{\sigma}$$

Remarks:

1. For small samples, the test is based on special tables.
2. This program can also be used for other tests involving runs. For example, one might want to test runs of scores above and below the median based on the order in which the scores were obtained. In this case, a sequence could be constructed in which each score would be replaced by a 1 if it was above the median or a 0, if below the median. The run test for randomness can then be applied to the sequence of 0's and 1's.

   Another use might be for Wald-Wolfowitz run test, which tests the null hypothesis that two random samples have been drawn from identical populations. The data from both groups are combined into one sequence according to magnitude. Each value may be assigned a 0 or 1 depending on which population it came from, and the run test for randomness then performed on the resulting sequence.

References: 1. Freund and Williams, Dictionary/Outline of Basic Statistics, McGraw-Hill, 1966.

2. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

A statistician sits by the roulette table one night in a Las Vegas casino, suspiciously watching the house rake in stake upon stake. To test the null hypothesis that the sequence of numbers is random, the statistician observes the following sequence of red (R) and black (B) numbers (ignoring 0 and 00):

RRRR B RRR BBBBB RR BBB RR BB RRR

In the sequence are 14 R's, 11 B's and a total of 9 runs. Find the mean and standard deviation of the sampling distribution and the z statistic.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] RUNTEST [ALPHA]	RUN TEST
	NO. OF RUNS?
9 [R/S]	NO. OF TYPE1?
14 [R/S]	NO. OF TYPE2?
11 [R/S]	MU=13.32
[R/S]	SIGMA=2.41
[R/S]	Z=-1.79

(His suspicion is not entirely unjustified).

User Instructions

				SIZE: 009
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program.			
2	Initialize the program.		[XEQ] RUNTEST	RUN TEST
				NO. OF RUNS?
3	Key in the number of runs.	u	[R/S]	NO. OF TYPE1?
4	Key in the number of type 1.	n_1	[R/S]	NO. OF TYPE2?
5	Key in the number of type 2.	n_2	[R/S]	MU=(μ)
			[R/S]	SIGMA=(σ)
			[R/S]	Z=(z)
6	For another case, go to step 2.			

Program Listings

Program	Comments
01♦LBL "RUN TEST"	
02 FIX 2	Initialize
03 "RUN TEST"	
04 AVIEW	
05 PSE	
06 "NO. OF RUNS ?"	
07 PROMPT	
08 STO 03	u
09 "NO. OF TYPE1?"	
10 PROMPT	
11 STO 01	n_1
12 "NO. OF TYPE2?"	
13 PROMPT	
14 STO 02	n_2
15 *	
16 2	
17 *	
18 STO 07	
19 RCL 01	
20 RCL 02	Calculate μ
21 +	
22 STO 08	
23 /	
24 1	
25 +	
26 STO 04	
27 "MU"	
28 XEQ 11	
29 RCL 07	
30 RCL 08	
31 -	
32 RCL 07	
33 *	
34 RCL 08	
35 ENTER↑	Calculate σ
36 *	
37 RCL 08	
38 1	
39 -	
40 *	
41 /	
42 SQRT	
43 STO 05	
44 "SIGMA"	
45 XEQ 11	
46 RCL 03	
47 RCL 04	
48 -	
49 RCL 05	Calculate z
50 /	
51 STO 06	
52 "Z"	
53♦LBL 11	
54 "⊢="	
55 ARCL X	
56 AVIEW	Display routine
57 STOP	
58 RTN	
59 .END.	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	n_1		
	n_2		
	u		
	μ		
05	σ	55	
	z		
	$2n_1n_2$		
	n_1+n_2		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 28 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF X

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

THE RUN TEST
FOR RANDOMNESS
PROGRAM REGISTERS NEEDED: 20

ROW 1 (1 - 2)

ROW 2 (3 - 6)

ROW 3 (6 - 7)

ROW 4 (8 - 9)

ROW 5 (9 - 12)

ROW 6 (12 - 20)

ROW 7 (21 - 29)

ROW 8 (30 - 42)

ROW 9 (43 - 48)

ROW 10 (49 - 57)

ROW 11 (58 - 59)

INTRACLASS CORRELATION COEFFICIENT

The intraclass correlation coefficient r_I measures the degree of association among individuals within classes or groups.

		Observations			
	1	x_{11}	x_{12}	...	x_{1n}
	2	x_{21}	x_{22}	...	x_{2n}
	.	.	.		.
Groups	.	.	.		.
	.	.	.		.
	k	x_{k1}	x_{k2}	...	x_{kn}

The coefficient is most easily calculated using the analysis of variance techniques. r_I is the sample estimate of the population intraclass correlation coefficient ρ_I. If we can assume that the individuals within groups are random samples from normal populations with the same variance, then the hypothesis $\rho_I = 0$ can be tested using the F statistic.

Equations:

1. Sums

   Group

$$T_i = \sum_{j=1}^{n} x_{ij} \qquad i = 1, 2, \ldots, k$$

   Total

$$T = \sum_{i=1}^{k} T_i$$

2. Sums of squares

   Mean

$$MSS = T^2/kn$$

   Among groups

$$ASS = \sum_{i=1}^{k} T_i{}^2/n - MSS$$

   Within groups

$$WSS = \sum_{i=1}^{k}\sum_{j=1}^{n} x_{ij}{}^2 - MSS - ASS$$

3. Intraclass correlation coefficient

$$r_I = \left(\frac{ASS}{k-1} - \frac{WSS}{k(n-1)}\right) \div \left(\frac{ASS}{k-1} + \frac{WSS}{k}\right)$$

4. F statistic

$$F = \frac{ASS}{k-1} \div \frac{WSS}{k(n-1)}$$

with $df_1 = k - 1$ and $df_2 = k(n - 1)$ degrees of freedom.

References: 1. B. Ostle, Statistics, in Research, Iowa State University Press, 1972.
2. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Groups		Observations	
	1	71	71
	2	69	72
	3	59	65
	4	65	64
	5	66	60
	6	73	72
	7	68	67
	8	70	68

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XRQ] [ALPHA] INT [ALPHA]	INTRACLASS C.
	N ?
2 [R/S]	X1,1 ?
71 [R/S]	X1,2 ?
71 [R/S]	T1=142
[R/S]	X2,1 ?
69 [R/S]	X2,2 ?
72 [R/S]	T2=141
⋮	⋮
70 [R/S]	X8,2 ?
68 [R/S]	T8=138
[E]	RI=0.70
[R/S]	F=5.61
[R/S]	dF1=7.00
[R/S]	dF2=8.00

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] INT	INTRACLASS C.
				N ?
3	Input n (the number of columns).	n	[R/S]	X1,1 ?
4	Perform steps 4-5 for i = 1, 2, ..., k	x_{ij}	[R/S]	X(i),(j+1)?
	and j = 1, 2, ..., n. T_i is automatically			Ti=(Ti)
	displayed when x_{in} is input. Press		[R/S]	X(i+1),I ?
	[R/S] to continue.			
5	Is you make a mistake inputting x_{ih},		[C]	X(i),(h)?
	correct it and go to step 4 (x_{in} cannot be	x_{ih} as entered	[R/S]	X(i),(h)?
	corrected -- go to step 2).			
6	Calculate r_I,		[E]	RI=(r_I)
	F,		[R/S]	F=(F)
	and the degrees of freedom.		[R/S]	dF1=(df_1)
			[R/S]	dF2=(df_2)
7	For another set of data, go to step 2.			

Program Listings

```
01♦LBL "INT"
02 FIX 0
03 CLRG
04 CF 29                  Initialize
05 CF 00
06 "INTRACLASS C."
07 AVIEW
08 PSE
09 "N ?"
10 PROMPT
11 STO 09
12 GTO a
- - - - - - - - - - - - - - - - - - -
13♦LBL C                  Correction
14 SF 00                  routine
15 1
16 ST- 01
- - - - - - - - - - - - - - - - - - -
17♦LBL a
18 RCL 01
19 1
20 +
21 RCL 02                 Input prompt
22 1                      routine
23 +
24 "X"
25 ARCL X
26 "├,"
27 ARCL Y
28 "├ ?"
29 PROMPT
- - - - - - - - - - - - - - - - - - -
30 FS? 00
31 CHS
32 ST+ 06
33 X↑2
34 FS? 00
35 CHS
36 ST+ 05
37 1
38 FC?C 00
39 ST+ 01
40 RCL 09                 n
41 RCL 01                 j
42 X≠Y?
43 GTO a
44 RCL 06
45 STO 08                 Calculate Ti
46 ST+ 03
47 X↑2
48 ST+ 04
49 0
50 STO 01
51 STO 06
52 1
53 ST+ 02
54 RCL 08
55 "T"
56 ARCL 02
57 XEQ 11
58 GTO a
- - - - - - - - - - - - - - - - - - -
59♦LBL E
60 FIX 2
61 RCL 04
62 RCL 03
63 X↑2
64 RCL 02
65 /
66 -
67 RCL 09
68 STO 01                 ASS
69 /
70 RCL 02                 Calculate rI
71 1
72 -
73 /
74 STO 00
75 RCL 05
76 RCL 04
77 RCL 01
78 /
79 -
80 RCL 02
81 /
82 STO 08                 WSS/k
83 RCL 01
84 1
85 -
86 STO 01
87 /
88 -
89 RCL 00
90 RCL 08
91 +
92 /
93 "RI"
94 XEQ 11
- - - - - - - - - - - - - - - - - - -
95 RCL 00
96 RCL 08
97 RCL 01                 Calculate F
98 /
99 /
100 "F"
- - - - - - - - - - - - - - - - - - -
```

Program Listings

```
101 XEQ 11
102 RCL 02
103 1
104 -
105 "dF1"
106 XEQ 11
107 RCL 01
108 RCL 02
109 *
110 "dF2"
111♦LBL 11
112 "⊢="
113 ARCL X
114 AVIEW
115 STOP
116 RTN
117 .END.
```

Calculate df_i

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	ASS/k-1	50	
	j → n		
	i → k		
	T		
	ΣT_i^2		
05	$\Sigma\Sigma x_{ij}^2$	55	
	Σx_{ij}		
	temp, WSS/k		
	n, n-1		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 37 USER MODE

ENG ___ FIX 2 SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		For corrections	Normal
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

INTRACLASS CORRELATION COEFFICIENT

PROGRAM REGISTERS NEEDED: 28

ROW 1 (1 - 5)

ROW 2 (5 - 6)

ROW 3 (6 - 12)

ROW 4 (13 - 21)

ROW 5 (22 - 28)

ROW 6 (28 - 36)

ROW 7 (36 - 44)

ROW 8 (45 - 54)

ROW 9 (55 - 60)

ROW 10 (60 - 72)

ROW 11 (73 - 85)

ROW 12 (86 - 94)

ROW 13 (95 - 104)

ROW 14 (105 - 110)

ROW 15 (110 - 117)

FISHER'S EXACT TEST FOR A 2 x 2 CONTINGENCY TABLE

Fisher's exact probability test is used for analyzing a 2 x 2 contingency table when the two independent samples are small in size.

a	b
c	d

Suppose a, b, c, d are the frequencies and a is the smallest frequency, this program calculates the following:

1. The exact probability p_0 of observing the given frequencies in a 2 x 2 table, when the marginal totals are regarded as fixed.
2. The exact probability p_i ($i = 1, 2, \ldots, a$) of each more extreme table having the same marginal totals.
3. The sum S_i of the probabilities of the first $i + 1$ tables.
4. The sum S of the probabilities of all tables with the same margins (i.e., $S = S_a$).

Equations:

1.

$$p_0 = \frac{(a + b)!(c + d)!(a + c)!(b + d)!}{N!a!b!c!d!}$$

where

$$N = a + b + c + d.$$

2. For the more extreme table (with the same margins)

a − i	b + i
c + i	d − i

$$p_i = \frac{(a + b)!(c + d)!(a + c)!(b + d)!}{N!(a - i)!(b + i)!(c + i)!(d - i)!}$$

where

i can be 1, 2, ... or a.

3.

$$S_n = \sum_{i=0}^{n} p_i$$

where

n can be 1, 2, ..., a.

4.

$$S = \sum_{i=0}^{a} p_i$$

Remarks: 1. a must be the smallest among the frequencies. Rearrange the table if necessary.

2. This program requires $N \leqslant 69$. However, Fisher's exact test is normally used for $N \leqslant 30$.

References: 1. S. Siegel, Nonparametric Statistics, McGraw-Hill, 1956.

2. Sir R. A. Fisher, Statistical Methods for Research Workers, Oliver and Boyd, 1950.

3. This program is a translation of the HP-65 Stat Pac 2 program.

Example:

Calculate p_0, p_1, p_2, S_4 and S for the following table

7	10
8	5

Note:

The table must be rearranged as

5	8
10	7

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] FIS [ALPHA]	FISHERS TEST
	a?
5 [R/S]	b?
8 [R/S]	c?
10 [R/S]	d?
7 [R/S]	P0=0.16
[A]	P1=0.06
[A]	P2=0.01
[A] [A] [R/S]	S4=0.23
[E]	S=0.23

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] FIS	FISHERS TEST
				a?
3	Input frequencies and calculate P.	a	[R/S]	b?
		b	[R/S]	c?
		c	[R/S]	d?
		d	[R/S]	PO=(P_o)
4	(Optional) Perform steps 4-5 for			
	i = 1, 2, ..., a. Calculate P_i.		[A]	Pi=(P_i)
5	Calculate S_i.		[R/S]	Si=(S_i)
6	Calculate the sum of all probabilities.		[E]	S=(S)
7	For another set of data, go to step 2.			

Program Listings

Initialize

```
01♦LBL "FIS
"
02 FIX 2
03 CF 01
04 CF 29
05 "FISHERS
TEST"
06 AVIEW
07 PSE
08 CLRG
```

Store a, b, c, d and calculate numerator of P_i

```
09 "a?"
10 PROMPT
11 STO 01
12 STO 08
13 "b?"
14 PROMPT
15 STO 02
16 +
17 STO 05
18 "c?"
19 PROMPT
20 STO 03
21 "d?"
22 PROMPT
23 STO 04
24 +
25 STO 06
26 FACT
27 RCL 05
28 FACT
29 *
30 RCL 05
31 RCL 06
32 +
33 FACT
34 /
35 RCL 01
36 RCL 03
37 +
38 FACT
39 *
40 RCL 02
41 RCL 04
42 +
43 FACT
44 *
45 STO 07
46 0
47 STO 05
48 RDN
```

Loop to calculate P_i

```
49♦LBL 00
50 RCL 01
51 FACT
52 /
53 RCL 02
54 FACT
55 /
56 RCL 03
57 FACT
58 /
59 RCL 04
60 FACT
61 /
62 ST+ 05
63 FS? 01
64 RTN
65 "P"
66 XEQ 11
```

Display S_i

```
67 RCL 05
68 "S"
69 XEQ 11
70 STOP
```

Set up to calculate P_{i+1}

```
71♦LBL A
72 1
73 ST- 01
74 ST+ 02
75 ST+ 03
76 ST- 04
77 ST- 08
78 ST+ 00
79 RCL 07
80 GTO 00
```

Calculate S

```
81♦LBL E
82 SF 01
83 RCL 08
84 0
85 X=Y?
86 XEQ 01
87 XEQ A
88 GTO E
```

Display S

```
89♦LBL 01
90 CF 01
91 RCL 05
92 "S="
93 ARCL X
94 AVIEW
95 STOP
```

Display routine

```
96♦LBL 11
97 FIX 0
98 ARCL 00
99 "⊢="
100 FIX 2
```

Program Listings

```
101 ARCL X
102 AVIEW
103 STOP
104 RTN
105 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	i	50	
	a		
	b		
	c		
	d		
05	a+b, S_n	55	
	c+d		
	numerator of P_i		
	a		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 33 USER MODE
ENG ____ FIX 2 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		Calculate S	Normal
29		Cleared for proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FISHER'S EXACT TEST FOR A 2 X 2 CONTINGENCY TABLE

PROGRAM REGISTERS NEEDED: 25

ROW 1 (1 - 4)

ROW 2 (5 - 5)

ROW 3 (6 - 14)

ROW 4 (15 - 23)

ROW 5 (24 - 36)

ROW 6 (37 - 49)

ROW 7 (50 - 62)

ROW 8 (62 - 69)

ROW 9 (69 - 76)

ROW 10 (76 - 83)

ROW 11 (84 - 90)

ROW 12 (90 - 98)

ROW 13 (98 - 105)

RCW 14 (105 - 105)

BARTLETT'S CHI-SQUARE STATISTIC

$$\chi^2 = \frac{f \ln s^2 - \sum_{i=1}^{k} f_i \ln s_i^2}{1 + \frac{1}{3(k-1)}\left[\left(\sum_{i=1}^{k} \frac{1}{f_i}\right) - \frac{1}{f}\right]}$$

where: s_i^2 = sample variance of the ith sample

f_i = degrees of freedom associated s_i^2

i = 1, 2, ..., k

k = number of samples

$$s^2 = \frac{\sum_{i=1}^{k} f_i s_i^2}{f}$$

$$f = \sum_{i=1}^{k} f_i$$

This χ^2 has a chi-square distribution (approximately) with k - 1 degrees of freedom which can be used to test the null hypothesis that s_1^2, s_2^2, ..., s_k^2 are all estimates of the same population variance σ^2; i.e., H_0: Each of s_1^2, s_2^2, ..., s_k^2 is an estimate of σ^2.

References: 1. Statistical Theory with Engineering Applications, A. Hald, John Wiley and Sons, 1960.

2. This program is a translation of the HP-65 Stat Pac 1 program.

Example:

Apply the program to the following data:

i	1	2	3	4	5	6
s_i^2	5.5	5.1	5.2	4.7	4.8	4.3
f_i	10	20	17	18	8	15

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] BAR [ALPHA]	BARTLETTS
	F1?
10 [R/S]	S1 SQ?
5.5 [R/S]	F2?
⋮	⋮
15 [R/S]	S6SQ?
4.3 [R/S]	F7?
[E]	CHI SQ=0.25
[R/S]	dF=5.00

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] BAR	BARTLETTS
3	Perform steps 3-4 for i = 1, 2, ..., k.			F1?
	Input f_i.	f_i	[R/S]	S(i) SQ?
	Input S_i^2.	S_i^2	[R/S]	F(i+1)?
4	If you make a mistake inputting f_h or			
	S_h^2, perform this step and go back to step 3.	f_h or S_h^2 as entered	[C]	F(h)? or S(h) SQ?
5	Calculate χ^2		[E]	CHI SQ=(χ^2)
	and df.		[R/S]	dF=(df)
6	To use the program for another set of			
	data, go to step 2.			

Program Listings

Initialize

```
01♦LBL "BAR
"
02 FIX 0
03 CLRG
04 CF 01
05 CF 29
06 "BARTLET
TS"
07 AVIEW
08 PSE
09 GTO A
```

Correct s_i^2

```
10♦LBL C
11 FS? 01
12 GTO 01
13 STO 08
14 RCL 01
15 *
16 ST- 00
17 RCL 08
18 LN
19 RCL 01
20 *
21 ST- 06
```

Correct f_i

```
22 1
23 ST- 05
24 GTO b
25♦LBL 01
26 ST- 03
27 1/X
28 ST- 04
```

Accumulate sums

```
29♦LBL A
30 "F"
31 RCL 05
32 1
33 +
34 ARCL X
35 "⊢?"
36 PROMPT
37 SF 01
38 STO 01
39 ST+ 03
40 1/X
41 ST+ 04
42♦LBL b
43 "S"
44 RCL 05
45 1
46 +
47 ARCL X
48 "⊢ SQ?"
49 PROMPT
50 CF 01
51 STO 08
52 RCL 01
53 *
54 ST+ 00
55 RCL 08
56 LN
57 RCL 01
58 *
59 ST+ 06
60 1
61 ST+ 05
62 GTO A
```

Calculate χ^2 and df

```
63♦LBL E
64 FIX 2
65 RCL 00
66 RCL 03
67 /
68 LN
69 RCL 03
70 *
71 RCL 06
72 -
73 RCL 04
74 RCL 03
75 1/X
76 -
77 RCL 05
78 1
79 -
80 STO 02
81 3
82 *
83 /
84 1
85 +
86 /
87 "CHI SQ"
88 XEQ 11
89 RCL 02
90 "dF"
```

Display routine

```
91♦LBL 11
92 "⊢="
93 ARCL X
94 AVIEW
95 STOP
96 RTN
97 .END.
```

00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	$\Sigma f_i s_i^2$	50	
	f_i		
	df		
	Σf_i		
	$\Sigma(1/f_i)$		
05	k	55	
	$\Sigma f_i \ln s_i^2$		
	s_i^2		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 32 USER MODE
ENG ____ FIX 2 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		Correct f_i	Correct s_i^2
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BARTLETT'S CHI-SQUARE
STATISTIC
PROGRAM REGISTERS NEEDED: 24

ROW 1 (1 - 5)

ROW 2 (5 - 8)

ROW 3 (9 - 16)

ROW 4 (16 - 24)

ROW 5 (25 - 33)

ROW 6 (34 - 41)

ROW 7 (41 - 48)

ROW 8 (48 - 56)

ROW 9 (57 - 64)

ROW 10 (64 - 76)

ROW 11 (77 - 87)

ROW 12 (87 - 92)

ROW 13 (92 - 97)

MANN-WHITNEY STATISTICS

This program calculates the Mann-Whitney test statistic on two independent samples of equal of unequal sizes. This test is designed for testing the null hypothesis of no difference between two populations.

Mann-Whitney test statistic is defined as:

$$U = n_1 \; n_2 + \frac{n_1 \; (n_1 + 1)}{2} - \sum_{i=1}^{n_1} R_i$$

where n_1 and n_2 are the sizes of the two samples and R_i (i = 1, 2, ..., n) is the rank assigned to the values of a given sample. All values from both samples should be arranged jointly (as if they were one sample) in an increasing order of magnitude.

When n_1 and n_2 are small, the Mann-Whitney test bases on the exact distribution of U and specially constructed tables. When n_1 and n_2 are both large (i.e., greater than 20) then:

$$Z = \frac{U - \frac{n_1 n_2}{2}}{\sqrt{n_1 \; n_2 (n_1 + n_2 + 1)/12}}$$

is approximately a random variable having the standard normal distribution.

If the size of neither sample is greater than 20, the user should consult the special U-tables (for example, Handbook of Statistical Tables, D. B. Owens, Addison-Wesley, 1962), using the smaller of the two possible U's (one for each sample). When this occurs, the program automatically determines and displays the approximate U and does not compute Z.

The following program includes two options. Option I assigns and enter ranks based on the number of times a datum occurs in both samples. Rank is determined by:

$$R_n = \frac{F_1 n + F_2 n + 1}{2} + \sum_{i=0}^{n-1} F_1 n + \sum_{i=0}^{n-1} F_2 n$$

Where $F_{10} = F_{20} = 0$

Frequencies are entered sequentially corresponding to increasingly larger data values. There is one error deletion routine for option I.

Option II is used when the ranks for the data values are already known. The inputs are the ranks and the corresponding frequencies for the sample. This option includes two error deletion routines.

References: 1. Mathematical Statistics, J. E. Freuno, Prentice-Hall, 1962.

2. Nonparametric Statistics for the Social Sciences, Sidney Siegel, McGraw-Hill, 1956, pp. 115-123; 271-277.

Find U and Z for the following data:

Example:

Sample 1		Sample 2	
Data	Ranks	Data	Ranks
4	4.5	4	4.5
4	4.5	4	4.5
4	4.5	4	4.5
		4	4.5
		4	4.5
6.2	10	6.2	10
6.2	10		
7.1	14.5	7.1	14.5
7.1	14.5	7.1	14.5
7.1	14.5	7.1	14.5
8	22.5	8	22.5
8	22.5	8	22.5
8	22.5	8	22.5
8	22.5	8	22.5
		8	22.5
		8	22.5
10	29	10	29
10	29		
		13	32
		13	32
		13	32
14	35	14	35
		14	35
17	37		

OPTION I (ranks not yet assigned):

i	Datum $Value_i$	$F1_i$	$F2_i$
1	4	3	5
2	6.2	2	1
3	7.1	3	3
4	8	4	6
5	10	2	1
6	13	0	3
7	14	1	2
8	17	1	0

OPTION II (ranks already assigned):

i	F_i	R_i
1	3	4.5
2	2	10
3	3	14.5
4	4	22.5
5	2	29
6	1	35
7	1	37

SOLUTION: Option I

Input	Function	Display	Comments
Load M-W	GTO..	Packing	Load program and set size
Set size 006			Start program
	[XEQ]M-W	Mann-Whitney	
		$1{:}F_1 \uparrow F_2?$	Enter the number of times a datum occurs in both samples
3	[ENTER]	3	
5	[R/S]	Rank = 4.5	
		$2{:}F_1 \uparrow F_2?$	
2	[ENTER]	2	
1	[R/S]	Rank = 10.0	
		$3{:}F_1 \uparrow F_2?$	
3	[ENTER]	3	
	[R/S]	Rank = 14.5	
		$4{:}F_1 \uparrow F_2?$	
44	[ENTER]	44	
66	[R/S]	Rank = 72.5	
		$5{:}F_1 \uparrow F_2?$	Oops! Need to correct that error.
	[XEQ] "a"	$4{:}F_1 \uparrow F_2?$	Input correct values & continue.
4	[ENTER]		
6	[R/S]	Rank = 22.5	
		$5{:}F_1 \uparrow F_2?$	
2	[ENTER]		
1	[R/S]	Rank = 29.0	
		$6{:}F_1 \uparrow F_2?$	
0	[ENTER]		
3	[R/S]	Rank = 32.0	
		$7{:}F_1 \uparrow F_2?$	
2	[R/S]	Rank = 35.0	1 is "entered" by default
		$8{:}F_1 \uparrow F_2?$	
0	[R/S]	Rank = 37.0	
		$9{:}F_1 \uparrow F_2?$	Last item already entered. Calculate U & Z.
	[XEQ]"C"	u=175.0000	
	[R/S]	z=0.2146	

OPTION II

Input	Function	*Display	Comments
Set size 007			
	[XEQ] "E"	Mann-Whitney	
		N_1?	No. data items - sample 1?
16	[R/S]	N_2?	No. data items - sample 2?
21	[R/S]	1:F↑R?	Enter frequency & rank
3	[ENTER]	3	
4.5	[R/S]	2:F↑R?	
3	[ENTER]	3	
100	[R/S]	3:F↑R?	Need to correct the last input
	[XEQ] "e"	3 100 deleted	
		2:F↑R?	Enter correct value
2	[ENTER]	2	
10	[R/S]	3:F↑R?	
3	[ENTER]	3	
14.5	[R/S]	4:F↑R?	
5	[ENTER]	5	
225	[R/S]	5:F↑R?	4 was entered incorrectly - to delete
2	[ENTER]	2	
29	[R/S]	6:F↑R?	
5	[ENTER]	5	
225	[XEQ] "d"	5 225 deleted	
		5:F↑R?	Enter correct value
4	[ENTER]	4	
22.5	[R/S]	6:F↑R?	
35	[R/S]	7:F↑R?	
37	[R/S]	U=175.0000	
	[R/S]	Z=0.2146	

* Display shown as appears without a printer - printer output shown on page # 55

PRINTER OUTPUT

Output I

```
MANN-WHITNEY

F1 = 3
F2 = 5
RANK = 4.5

F1 = 2
F2 = 1
RANK = 10.0

F1 = 3
F2 = 3
RANK = 14.5

F1 = 44
F2 = 66
RANK = 72.5

F1 = 44
F2 = 66

F1 = 4
F2 = 6
RANK = 22.5

F1 = 2
F2 = 1
RANK = 29.0

F1 = 0
F2 = 3
RANK = 32.0

F1 = 1
F2 = 2
RANK = 35.0

F1 = 1
F2 = 0
RANK = 37.0

U=175.0000
Z=0.2146
```

PRINTER OUTPUT

Output II

```
MANN-WHITNEY

N1 = 16
N2 = 21

F=3.0   R=4.5
F=3.0   R=100.0
3  100 DELETED
F=2.0   R=10.0
F=3.0   R=14.5
F=5.0   R=225.0
F=2.0   R=29.0
5  225 DELETED
F=4.0   R=22.5
F=1.0   R=35.0
F=1.0   R=37.0

U=175.0000
Z=0.2146
```

User Instructions

Option I (ranks not yet assigned)

SIZE: 006

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY**
I1	Load program*, pack, set size			
I2	Initialize (Option I)		[XEQ] "M-W"	Mann-Whitney 1:$F_1 \uparrow F_2$?
	Perform steps I3 & I4 for all data values i = 1,2.. n where n is the number of			
	different data values in both samples			
I3	Input number of times datum_i occurs			
	in sample 1. Datum_1 is the smallest valued			
	datum, datum_2 is the next larger, etc. If			
	F_{1i}=1 this step may be skipped	F_{1i}	[ENTER]	F_{1i}
I4	Input number of times datum_i occurs in			
	sample 2	F_{2i}	[R/S]	Rank = (Ri) (i+1):$F_1 \uparrow F_2$?
I5	If an error was made when inputting the			
	previous entry, delete the entry and go to			
	step I3		[XEQ] "a"	(i):$F_1 \uparrow F_2$?
I6	Calculate U		[XEQ] "c"	U= (a)
I7	Calculate Z if either sample size is			
	greater than 20		[R/S]	Z= (z)
	(*Note: If only Option I will be used, program lines 79 - 127 can be deleted)			
	**If printer is attached, all values will be printed (ie $F_1, \uparrow F_2$, rank, U, Z)			

User Instructions

Option II (ranks already assigned) SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY **
II1	Load program,* pack, set size 007			
II2	Initialize (Option II)		(GTO) "M-W" [XEQ] E	Mann-Whitney N1?
II3	Input number of data items in sample 1	N1	[R/S]	N2?
II4	Input number of data items in sample 2	N2	[R/S]	1:F↑R?
	Perform steps II5 & II6 for i=1,2...,n, where n is the number of different rank values in sample 1.			
II5	Input the number of times R_i occurs in sample 1. If F_i=1, this step may be skipped.	F_i	[ENTER]	(F_i)
II6	Input R_i	R_i	[R/S]	(i+1):F↑R?
II7	If a mistake was made in the pair of entries just entered, delete the pair and go to step II5		[XEQ]"e"	$F_i R_i$ deleted (i):F↑R?
II8	If a mistake was made when inputting a different pair of entries (ie, not the most recent) re-enter the incorrect pair, delet, go to step II 5	Fm	[ENTER]	(Fm)
		Rm	[XEQ]"d"	Fm Rm deleted (i):F↑R?
II9	U is calculated (automatically) after the nth input			U=(U)
II10	Z is calculated if either sample size is greater than 20.		[R/S]	Z=(Z)

*(Note if only Option II is to be used, program lines 02-60 can be deleted.)

**If a printer is present - all values will be printed.

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00 ΣR_i
01 N_1
02 N_2
03 Counter for data entry
04 Option I: F_{1i}
Option II: F_i
05 Option I: $F2_i$
Option II: R_i
06 No. of data processed (Option II only)

STATUS

SIZE 6/7* TOT. REG. 59/60** USER MODE
ENG ___ FIX 0,1,4 SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
12		Print double wide	Print single wide
21		Printer enable	Printer disable
27	S	User mode set	User mode off
29		Digit separators used in display	Digit separators omitted
55		Printer exists	No printer

* Option I requires size 006
Option II requires size 007
** Depending on size set (6 or 7)

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

01♦LBL "M-W"	
02♦LBL A	Begin Option I
03 XEQ 03	
04♦LBL 01	
05 1	Set up counter
06 ST+ 03	
07 CLA	
08 ARCL 03	
09 "⊢:F1↑F2?"	Prompt for input
10 PROMPT	
11 STO 05	Store F_{2i}
12 X<>Y	
13 STO 04	Store F_{1i}
14 XEQ 02	
15 FIX 1	
16 "RANK = "	
17 ARCL X	
18 AVIEW	
19 PSE	Display calculated
20 FIX 0	Rank (R_i)
21 RCL 04	
22 ST+ 01	Accumulate
23 *	F_{1i} ($\to n_1$)
24 ST+ 00	Accumulate R_i
25 RCL 05	
26 ST+ 02	Accumulate F_{2i}
27 GTO 01	($\to n_2$)
28♦LBL a	Subroutine for
29 RCL 05	deleting F_{1n} and
30 ST- 02	F_{2n}
31 RCL 04	
32 ST- 01	(Option I)
33 XEQ 02	
34 RCL 04	
35 *	
36 ST- 00	
37 2	
38 ST- 03	
39 GTO 01	
40♦LBL 02	Calculate Rank
41 +	
42 STO Y	
43 1	
44 +	
45 2	
46 /	
47 RCL 01	
48 RCL 02	
49 +	Printer Attached?
50 +	If not return (to
51 FC? 55	line #15)
52 RTN	
53 ADV	
54 "F1 = "	
55 ARCL 04	If attached echo
56 AVIEW	print F1 and F2
57 "F2 = "	
58 ARCL 05	
59 AVIEW	
60 RTN	
61♦LBL 03	
62 SF 12	
63 CF 21	Set flag 21 to
64 FS? 55	match status of
65 SF 21	flag 55
66 "MANN-WHITNEY"	Output header
67 AVIEW	
68 PSE	
69 CF 12	
70 0	
71 STO 00	Initialize neces-
72 STO 01	sary registers
73 STO 02	
74 STO 03	
75 SF 27	
76 CF 29	Set user mode
77 FIX 0	
78 RTN	
79♦LBL e	
80 RCL 04	Recall F_n & R_n for
81 RCL 05	deletion (OptionII)
82♦LBL d	Subroutine for
83 CLA	deletion of entries
84 ARCL Y	(Option II)
85 "⊢ "	
86 ARCL X	
87 "⊢ DELETED"	
88 AVIEW	
89 X<>Y	
90 ST- 06	
91 *	
92 ST- 00	
93 2	
94 ST- 03	
95 GTO 04	
96♦LBL E	
97 XEQ 03	Begin Option II
98 STO 06	Prompt for and
99 "N1?"	store N1 & N2
100 PROMPT	
101 STO 01	
102 "N2?"	

Program Listings

```
PROMPT
STO 02          If no printer
FC? 55          jump to Label 4
GTO 04
ADV             If printer exists
"N1 = "         display input for
ARCL 01         N1 & N2
AVIEW
"N2 = "
ARCL 02
AVIEW
ADV
♦LBL 04
1               Set up counter
ST+ 03
CLA
ARCL 03         Prompt for & store
"⊢:F↑R?"        input (Fi & Ri)
PROMPT
STO 05
X<>Y
STO 04
FS? 55          Printer exist?
XEQ 07          Jump to Label 7
ST+ 06          Number of data
*               Calculate RI &
ST+ 00          accumulate
RCL 01
RCL 06
X<Y?            Any more entries?
GTO 04
♦LBL C
SF 29
FIX 4
RCL 01          Compute u for
RCL 02          Sample 1
*               (Option I)
RCL 01
1
+
RCL 01
*
2
/
+
RCL 00
-
"U="            Determine if
20              sample size >20
RCL 01          If so calculate Z
X>Y?
GTO 05
CLX
```

```
RCL 02
X>Y?
GTO 05
RCL Z           Compute U for
STO Y           Sample 2
CHS
RCL 01
RCL 02
*
+
X>Y?            Select smaller u
X<>Y            & display (Sample
GTO 06          <20)
♦LBL 07         Subroutine to echo
FIX 1           print values of F
"F="            & R if printer
ARCL 04         attached
"⊢     R="
ARCL 05
AVIEW
FIX 0
RTN
♦LBL 05
SF 21
FS? 55
ADV             Display U
RCL Z           (Sample >20)
ARCL X
AVIEW
RCL 01          Calculate value
RCL 02          of Z
*
2
/
-
RCL 01
RCL 02
*
RCL 01
RCL 02
+
1
+
*
12
/
SQRT            Display final out-
/               put (U or Z depen-
"Z="            ding on sample
♦LBL 06         size)
ARCL X
AVIEW
.END.
```

MANN–WHITNEY STATISTIC

PROGRAM REGISTERS NEEDED: 54

ROW 1 (1 : 4)

ROW 2 (5 : 9)

ROW 3 (9 : 16)

ROW 4 (16 : 21)

ROW 5 (22 : 29)

ROW 6 (30 : 37)

ROW 7 (38 : 47)

ROW 8 (48 : 54)

ROW 9 (55 : 60)

ROW 10 (61 : 66)

ROW 11 (66 : 69)

ROW 12 (70 : 79)

ROW 13 (79 : 86)

ROW 14 (86 : 89)

ROW 15 (90 : 97)

ROW 16 (97 : 102)

ROW 17 (103 : 108)

ROW 18 (109 : 114)

MANN–WHITNEY STATISTIC

ROW 19 (115 : 120)

ROW 20 (120 : 128)

ROW 21 (129 : 136)

ROW 22 (137 : 149)

ROW 23 (150 : 158)

ROW 24 (158 : 168)

ROW 25 (168 : 173)

ROW 26 (173 : 180)

ROW 27 (180 : 190)

ROW 28 (191 : 202)

ROW 29 (203 : 208)

KENDALL'S COEFFICIENT OF CONCORDANCE

Suppose n individuals are ranked from 1 to n according to some specified characteristic by k observers, the coefficient of concordance W measures the agreement between observers (or concordance between rankings).

$$W = \frac{12 \sum_{i=1}^{n} \left(\sum_{j=1}^{k} R_{ij} \right)^2}{k^2 \, n(n^2 - 1)} - \frac{3(n + 1)}{n - 1}$$

Where R_{ij} is the rank assigned to the ith individual by the jth observer.

W varies from 0 (no community of preference) to 1 (perfect agreement). The null hypothesis that the observers have no community of preference may be tested using special tables, or if n > 7, by calculating

$$\chi^2 = k\ (n - 1)\ W$$

which has approximately the chi-aquare distribution with n - 1 degrees of freedom (df).

Operating Limits and Warnings:

For small samples (say, less than or equal to 7) the specially constructed tables should be used. For example: Rank Correlation Methods, M.G. Kendall, Hafner Publishing Co., 1962.

References: 1. Nonparametric Statistical Inference, J. D. Gibbond, McGraw-Hill, 1971.

2. This program is a translation of the HP-65 Stat Pac 1 program.

Example:

Find W, χ^2, and df for the following data:

Table for R_{ij} (n = 4, k = 3)

i \ j	1	2	3
1	6	7	3
2	1	4	2
3	9	3	5
4	2	6	1

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] KEN [ALPHA]	KENDALLS COF.
	K?
3 [R/S]	R1,1 ?
6 [R/S]	R1,2 ?
7 [R/S]	R1,3 ?
3 [R/S]	S1=16
[R/S]	R2,1 ?
1 [R/S]	R2,2 ?
⋮	⋮
⋮	R4,3 ?
1 [R/S]	S4=9
[E]	W=10.00
[R/S]	CHI SQ=90.00
[R/S]	dF=3.00

NOTE: Although this example violates the warning ($n < 7$), the amount of data to be entered has been kept small to allow the user to run through the example in short order.

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode.		[USER]	
2	Initialize the program.		[XEQ] KEN	KENDALLS COF.
				K?
3	Input k.	k	[R/S]	R1,1 ?
4	Input R_{ij}. Repeat steps 4-5 for			
	i = 1, 2, ..., k.	R_{ij}	[R/S]	R(i),(j+1)
5	If you make a mistake inputting R_{ih},			
	delete it and go to step 4.	R_{ih}	[C]	R(i),(h)?
6	The sum of the i'th row is automatically			Si=(ΣR_{ij})
	calculated when $R_{i,k}$ is input. Press		[R/S]	R(i+1),1 ?
	[R/S] to continue, or calculate W,		[E]	W=(W)
	χ^2,		[R/S]	CHI SQ=(χ^2)
	and df.		[R/S]	dF=(df)
7	For another set of data, go to step 2.			

Program Listings

Initialize

```
01◆LBL "KEN"
02 CLRG
03 FIX 0
04 CF 29
05 "KENDALLS COF."
06 AVIEW
07 PSE
08 "K?"
09 PROMPT
10 STO 05
11 GTO A
```

Correction routine

```
12◆LBL C
13 ST- 02
14 1
15 ST- 01
```

Accumulate sums

```
16◆LBL A
17 "R"
18 RCL 01
19 1
20 +
21 RCL 04
22 1
23 +
24 ARCL X
25 "⊢,"
26 ARCL Y
27 "⊢ ?"
28 PROMPT
29 ST+ 02
30 1
31 ST+ 01
32 RCL 01
33 RCL 05
34 X>Y?
35 GTO A
36 1
37 ST+ 04
38 RCL 02
39 STO 06
40 X↑2
41 ST+ 03
42 0
43 STO 01
44 STO 02
45 RCL 06
46 "S"
47 "⊢"
48 ARCL 04
49 XEQ 11
50 GTO A
```

Calculate W

```
51◆LBL E
52 FIX 2
53 RCL 03
54 12
55 *
56 RCL 05
57 X↑2
58 /
59 RCL 04
60 /
61 RCL 04
62 X↑2
63 1
64 -
65 /
66 RCL 04
67 1
68 ST- 04
69 +
70 3
71 *
72 RCL 04
73 /
74 -
75 "W"
76 XEQ 11
```

Calculate χ^2 and df

```
77 RCL 05
78 *
79 RCL 04
80 *
81 "CHI SQ"
82 XEQ 11
83 RCL 04
84 "dF"
```

Display routine

```
85◆LBL 11
86 "⊢="
87 ARCL X
88 AVIEW
89 STOP
90 RTN
91 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	j ... k		
	ΣR_{ij}		
	$\Sigma (R_{ij})^2$		
	i ... n		
05	K	55	
	ΣR_{ij}		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 30 USER MODE
ENG ___ FIX 2 SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
29		For proper display format	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

KENDALL'S COEFFICIENT OF CONCORDANCE

PROGRAM REGISTERS NEEDED: 24

ROW 1 (1 - 5)

ROW 2 (5 - 5)

ROW 3 (6 - 13)

ROW 4 (13 - 22)

ROW 5 (23 - 28)

ROW 6 (29 - 37)

ROW 7 (37 - 47)

ROW 8 (47 - 52)

ROW 9 (53 - 64)

ROW 10 (65 - 75)

ROW 11 (76 - 81)

ROW 12 (81 - 87)

ROW 13 (87 - 91)

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41 programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41, HP has an extensive library of "Application Pacs". These programs transform your HP-41 into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, miniature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41.

You can choose from:

Aviation (Pre-Flight Only) 00041-15018
Clinical Lab 00041-15024
Circuit Analysis 00041-15024
Financial Decisions 00041-15004
Mathematics 00041-15003
Structural Analysis 00041-15021
Surveying 00041-15005
Securities 00041-15026
Statistics 00041-15002
Stress Analysis 00041-15027
Games 00041-15022
Home Management 00041-15023
Machine Design 00041-15020
Navigation 00041-15017
Real Estate 00041-15016
Thermal and Transport Science 00041-15019
Petroleum Fluids 00041-15039

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41 Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales 00041-90094
Home Construction Estimating 00041-90096
Lending, Saving and Leasing 00041-90086
Real Estate 00041-90136
Small Business 00041-90137
Geometry 00041-90084
High-Level Math 00041-90083
Test Statistics 00041-90082
Antennas 00041-90093
Chemical Engineering 00041-90100
Control Systems 00041-90092
Electrical Engineering 00041-90088
Fluid Dynamics and Hydraulics 00041-90139
Games II 00041-90443
Civil Engineering 00041-90089
Heating, Ventilating & Air Conditioning 00041-90140
Mechanical Engineering 00041-90090
Solar Engineering 00041-90138
Calendars 00041-90145
Cardiac/Pulmonary 00041-90097
Chemistry 00041-90102
Games 00041-90099
Optometry I (General) 00041-90143
Optometry II (Contact Lens) 00041-90144
Physics 00041-90142
Surveying 00041-90141
Time Module Solutions 00041-90395

*Some books require additional memory modules to accomodate all programs.

TEST STATISTICS

ONE SAMPLE TEST STATISTICS FOR THE MEAN
TEST STATISTICS FOR THE CORRELATION COEFFICIENT
DIFFERENCES AMONG PROPORTIONS
BEHRENS-FISHER STATISTIC
KRUSKAL-WALLIS STATISTIC
MEAN-SQUARE SUCCESSIVE DIFFERENCE
THE RUN TEST FOR RANDOMNESS
INTRACLASS CORRELATION COEFFICIENT
FISHER'S EXACT TEST FOR A 2 × 2 CONTINGENCY TABLE
BARTLETT'S CHI-SQUARE STATISTIC
MANN-WHITNEY STATISTIC
KENDALL'S COEFFICIENT OF CONCORDANCE

HEWLETT PACKARD

00041-90082 Rev B 7/82 Printed in U.S.A.